Ehlers MESA Adaptive Moving Averages (MAMA & FAMA)

everget WIZARD Eyl 20, 2018

Trend Analysis · Moving Averages · Exponential Moving Average (EMA) · ehlers · mesa · adaptive · Adaptive Moving Average (AMA) · MESA Adaptive Moving Average (MAMA) · fama · waveform · hilbert · phase

20 · 1068

Eyl 20, 2018 — Ehlers MESA Adaptive Moving Averages (MAMA & FAMA) script.

These indicators was originally developed by John F. Ehlers (Stocks & Commodities V. 19:10: MESA Adaptive Moving Averages).

Eyl 21, 2018 — **Sürüm Notları**: Added missed radians to degrees conversion. Thanks to @sal157011

Eyl 21, 2018 — **Sürüm Notları**: Added ribbon filling

Eyl 22, 2018 — **Sürüm Notları**: Refactored

Ara 12, 2018 — **Sürüm Notları**:
- Fixed issue with decimal step in inputs

Oca 31, 2019 — **Sürüm Notları**:
- Refactored

Şub 10, 2019 — **Sürüm Notları**:
- Fixed error and refactored

Eyl 15, 2019 — **Sürüm Notları**:
- Converted to v4
- Added alerts

Eyl 25, 2020 — **Sürüm Notları**:
- Update

Freelance -> Telegram: @alex_everget

A list of Free indicators:
https://bit.ly/2S7EPuN

A list of Paid indicators:
https://bit.ly/33MA81f

Earn $30:
https://www.tradingview.com/gopro/?share_your_love=everget

Website

Açık kaynak kodlu komut dosyası

Gerçek TradingView ruhuyla, bu betiğin yazarı, yatırımcının anlayabilmesi ve doğrulayabilmesi için onu açık kaynak olarak yayınladı. Yazarın eline sağlık! Bunu ücretsiz olarak kullanabilirsiniz, ancak bu kodun bir yayında yeniden kullanımı Kullanım Koşulları ile yönetilir. Bir grafikte kullanmak için favorilere ekleyebilirsiniz.

Feragatname

Bilgiler ve yayınlar, TradingView tarafından sağlanan veya onaylanan finansal, yatırım, işlem veya diğer türden tavsiye veya tavsiyeler anlamına gelmez ve teşkil etmez. Kullanım Şartları'nda daha fazlasını okuyun.

Bu komut dosyasını bir grafikte kullanmak ister misiniz?

★ Favori göstergelere ekle

```
//@version=4
// Copyright (c) 2018-present, Alex Orekhov (everget)
// Ehlers MESA Adaptive Moving Averages (MAMA & FAMA)script may be freely distributed under the terms of the GPL-3.0 license.
study("Ehlers MESA Adaptive Moving Averages (MAMA & FAMA)", shorttitle="MAMA & FAMA", overlay=true)

fastLimit = input(title="Fast Limit", type=input.float, step=0.01, defval=0.5)
slowLimit = input(title="Slow Limit", type=input.float, step=0.01, defval=0.05)
applyFilling = input(title="Apply Ribbon Filling ?", type=input.bool, defval=true)
src = input(title="Source", type=input.source, defval=hl2)

var float PI = 2 * asin(1)

// Truncated Hilbert transform
_hilbertTransform(src) =>
    out= 0.0962 * src + 0.5769 * nz(src[2]) - 0.5769 * nz(src[4]) - 0.0962 * nz(src[6])
    out

_computeComponent(src, mesaPeriodMult) =>
```

```
    out = _hilbertTransform(src) * mesaPeriodMult
    out

_smoothComponent(src) =>
    out = 0.2 * src + 0.8 * nz(src[1])
    out

_computeAlpha(src, fastLimit, slowLimit) =>
    mesaPeriod = 0.0
    mesaPeriodMult = 0.075 * nz(mesaPeriod[1]) + 0.54

    smooth = (4 * src + 3 * nz(src[1]) + 2 * nz(src[2]) + nz(src[3])) / 10
    detrender = _computeComponent(smooth, mesaPeriodMult)

    // Compute InPhase and Quadrature components
    I1 = nz(detrender[3])
    Q1 = _computeComponent(detrender, mesaPeriodMult)

    // Advance the phase of I1 and Q1 by 90 degrees
    jI = _computeComponent(I1, mesaPeriodMult)
    jQ = _computeComponent(Q1, mesaPeriodMult)

    // Phasor addition for 3 bar averaging
    I2 = I1 - jQ
    Q2 = Q1 + jI

    // Smooth the I and Q components before applying the discriminator
    I2 := _smoothComponent(I2)
    Q2 := _smoothComponent(Q2)

    // Homodyne Discriminator
    Re = I2 * nz(I2[1], I2) + Q2 * nz(Q2[1], Q2)
    Im = I2 * nz(Q2[1], Q2) - Q2 * nz(I2[1], I2)

    Re := _smoothComponent(Re)
    Im := _smoothComponent(Im)

    if Re != 0 and Im != 0
        mesaPeriod := 2 * PI / atan(Im / Re)

    mesaPeriod := min(mesaPeriod, 1.5 * nz(mesaPeriod[1], mesaPeriod))
    mesaPeriod := max(mesaPeriod, 0.67 * nz(mesaPeriod[1], mesaPeriod))
    mesaPeriod := min(max(mesaPeriod, 6), 50)
    mesaPeriod := _smoothComponent(mesaPeriod)

    phase = 0.0

    if I1 != 0
        phase := (180 / PI) * atan(Q1 / I1)

    deltaPhase = nz(phase[1], phase) - phase
    deltaPhase := max(deltaPhase, 1)

    alpha = max(fastLimit / deltaPhase, slowLimit)

    out = alpha
    out

alpha = _computeAlpha(src, fastLimit, slowLimit)
alpha2 = alpha / 2

mama = 0.0
mama := alpha * src + (1 - alpha) * nz(mama[1])

fama = 0.0
fama := alpha2 * mama + (1 - alpha2) * nz(fama[1])

mamaPlot = plot(mama, title="MAMA", linewidth=2, color=#674ea7, transp=0)
famaPlot = plot(fama, title="FAMA", linewidth=2, color=#f6b26b, transp=0)

noneColor = color.new(color.white, 100)

fillColor = applyFilling ? (mama > fama ? #0ebb23 : #cc0000) : noneColor
fill(mamaPlot, famaPlot, color=fillColor, transp=80)

long = crossover(mama, fama)
short = crossunder(mama, fama)

alertcondition(long, title="Alert: Long", message="MAMA & FAMA Long!")
alertcondition(short, title="Alert: Short", message="MAMA & FAMA Short!")

```

Yorumlar

Yararlı veya teşvik edici bir yorum bırakın. Piyasalara birlikte hakim olalım

Alkışlarla yorum | Yorum Paylaş

sal157011 · Eyl 21, 2018

The code is wrong because in Tradestation the function atan() returns degrees while in PineScript returns radians.
To fix the problem the result of the function in degrees must be converted into radians multiplying it by 180/pi.
You must create de variable pi= 3.14159265359 first and
line 50 mesaPeriod:= 360/atan(Im/Re) must be rewritten into mesaPeriod:= 2*pi/atan(Im/Re)
line 72 phase:=atan(Q1/I1) must be rewritten into phase:= 180/pi * atan(Q1 / I1)

note: 360° = 2*pi radians

The chart should look something like this



+6 Cevap Gönder

everget WİZARD · Eyl 21, 2018

@sal157011, thank you. My bad.

+1 Cevap Gönder

janor123 PREMİUM · Nis 22, 2020

How would you compare this to the JMA (jurik moving average)? Have you tested/compared it?

+2 Cevap Gönder

everget WİZARD · Nis 23, 2020

@janor123, yup

Cevap Gönder

janor123 PREMİUM · Eki 22, 2020

@everget, which one seemed better?

+1 Cevap Gönder

everget WİZARD · Eki 22, 2020

@janor123, I have a backtester for that



Over 60 different types of moving averages are available

+1 Cevap Gönder

Kunzat · Eyl 26, 2020

thnax 4 update!

+1 Cevap Gönder

everget WİZARD · Eyl 26, 2020

@Kunzat, you're welcome!

Cevap Gönder

blackcat1402 PREMİUM · Eyl 13, 2020

thanks for sharing this

+1 Cevap Gönder

everget WİZARD · Eyl 26, 2020

@blackcat1402, you're welcome

Cevap Gönder

Acel PRO · Eyl 21, 2018

This is awesome!

I was thinking of working on this lately. I want to experiment with different alphas on the EMA.

+1 Cevap Gönder

Neverstorm · Eyl 20, 2018

Thanks!! Youre really on a roll releasing all these quality Alogrithms :)
Keep up the good work!

+1 Cevap Gönder

everget WİZARD · Eyl 20, 2018

@Neverstorm, okay)

+3 Cevap Gönder

virenchocha · Eyl 18, 2021

Fortunately, your own indicator Half Trend is already beating this indicator. In fact , Half trend is sort of benchmark against I compare other indicators.

Cevap Gönder

jeffreyleo · Tem 12, 2021

I noticed a problem with how your smoothComponent() works vs the inline code it replaces : for example Q2 := _smoothComponent(Q2) vs Q2 := .2*Q2 + .8nz(Q2)
Since it is self-updating, the Q2 outside the function refers to the Q2 value after the function updates it but the src/Q2 inside the function refers to the Q2 value before it's been updated.
It does make a big difference in dominant cycle period result.

Cevap Gönder

jeffreyleo · Tem 12, 2021

@jeffreyleo, I am specifically referring to the value of the history variable Q2 out the function vs src inside the function being the problem (typo above Q2 := .2*Q2 + .8nz(Q2)

Cevap Gönder

theforexminer PREMİUM · Nis 28, 2020

I guess my comment/question really won't apply because according to other comments, it is not working correctly. Has anyone updated it to working correctly status? Thank you

Cevap Gönder

everget WİZARD · Nis 29, 2020

@forexmike, what?

Cevap Gönder

theforexminer PREMİUM · Nis 28, 2020

Hello, Do I just copy and paste this code into trading view to create and use the indicator? Thanks, Mike

Cevap Gönder

everget WİZARD · Nis 29, 2020

@forexmike, you can find the indicator through "Indicators & Strategies" button on your chart

Cevap Gönder