Ehlers Instantaneous Trendline

everget WIZARD Eki 7, 2018

Trend Analysis · Moving Averages · Exponential Moving Average (EMA) · ehlers · dominant · cycle · instantaneous · inphase · quadrature · hilbert · adaptive

9 · 566

Eki 7, 2018 — Ehlers Instantaneous Trendline script.

This indicator was described by John F. Ehlers in his book "Rocket Science for Traders" (2001, Chapter 10: The Instantaneous Trendline).

Eki 20, 2018 — **Sürüm Notları**: Refactored

Şub 10, 2019 — **Sürüm Notları**:
- Fixed error and refactored

Haz 27, 2019 — **Sürüm Notları**:
- Added ribbon filling

Freelance -> Telegram: @alex_everget

A list of Free indicators:
https://bit.ly/2S7EPuN

A list of Paid indicators:
https://bit.ly/33MA81f

Earn $30:
https://www.tradingview.com/gopro/?share_your_love=everget

Website


```
//@version=3
// Copyright (c) 2018-present, Alex Orekhov (everget)
// Ehlers Instantaneous Trendline script may be freely distributed under the MIT license.
study("Ehlers Instantaneous Trendline", shorttitle="EIT", overlay=true)

src = input(title="Source", type=source, defval=close)
applyFilling = input(title="Apply Ribbon Filling ?", type=bool, defval=true)

// Truncated Hilbert transform
hilbertTransform(src) =>
    0.0962 * src + 0.5769 * nz(src[2]) - 0.5769 * nz(src[4]) - 0.0962 * nz(src[6])

computeComponent(src, periodMult) =>
    hilbertTransform(src) * periodMult

computePart(src) =>
    0.2 * src + 0.8 * nz(src[1])

// FIR Filter
_fir(src) =>
    (4 * src + 3 * nz(src[1]) + 2 * nz(src[2]) + nz(src[3])) / 10

fir = _fir(src)

_eit(src) =>
    PI = 2 * asin(1)

    mesaPeriod = 0.0
    mesaPeriodMult = 0.075 * nz(mesaPeriod[1]) + 0.54

    detrender = 0.0
    detrender := computeComponent(fir, mesaPeriodMult)

    // Compute InPhase and Quadrature components
    I1 = nz(detrender[3])
    Q1 = computeComponent(detrender, mesaPeriodMult)
```

```
    // Advance the phase of I1 and Q1 by 90 degrees
    jI = computeComponent(I1, mesaPeriodMult)
    jQ = computeComponent(Q1, mesaPeriodMult)

    I2 = 0.0
    Q2 = 0.0

    // Phasor addition for 3 bar averaging
    I2 := I1 - jQ
    Q2 := Q1 + jI

    // Smooth the I and Q components before applying the discriminator
    I2 := computePart(I2)
    Q2 := computePart(Q2)

    // Homodyne Discriminator
    Re = I2 * nz(I2[1]) + Q2 * nz(Q2[1])
    Im = I2 * nz(Q2[1]) - Q2 * nz(I2[1])

    Re := computePart(Re)
    Im := computePart(Im)

    if Re != 0 and Im != 0
        mesaPeriod := 2 * PI / atan(Im / Re)

    if mesaPeriod > 1.5 * nz(mesaPeriod[1])
        mesaPeriod := 1.5 * nz(mesaPeriod[1])

    if mesaPeriod < 0.67 * nz(mesaPeriod[1])
        mesaPeriod := 0.67 * nz(mesaPeriod[1])

    if mesaPeriod < 6
        mesaPeriod := 6

    if mesaPeriod > 50
        mesaPeriod := 50

    mesaPeriod := 0.2 * mesaPeriod + 0.8 * nz(mesaPeriod[1])

    smoothPeriod = 0.0
    smoothPeriod := 0.33 * mesaPeriod + 0.67 * nz(smoothPeriod[1])

    // Compute Trendline as a SMA over the measured dominant cycle period
    dcPeriod = floor(smoothPeriod + 0.5)

    if dcPeriod < 1
        dcPeriod := 1

    itrend = 0.0

    for i = 0 to dcPeriod - 1
        itrend := itrend + src[i]

    if dcPeriod > 0
        itrend := itrend / dcPeriod

    eit = _fir(itrend)

    if n < 12
        eit := src
    eit

eit = _eit(src)

eitPlot = plot(eit, title="Trendline", linewidth=2, color=orange, transp=0)
firPlot = plot(fir, title="Price", linewidth=2, color=#6d1e7f, transp=0)

fillColor = applyFilling ? (fir > eit ? #0ebb23 : #cc0000) : color(white, 100)
fill(firPlot, eitPlot, color=fillColor, transp=80)
```

Yorumlar

Yararlı veya teşvik edici bir yorum bırakın. Piyasalara birlikte hakim olalım

Alkışlarla yorum | Yorum Paylaş

Saopaulo2010 · Ağu 31, 2020

This indicator is great!.. i would love to see this with multitime frame, so you sould for example see the 1 hr instantaneous trend but with candlestick for maybe 5 minutes, you could not miss a trade with that . Could you add the option?

+3 Cevap Gönder

everget WİZARD · Ağu 31, 2020

@Saopaulo2010, Hi, I can create a special version for you. Let me know in DM please

+1 Cevap Gönder

Saopaulo2010 · Eyl 9, 2020

@everget, What is DM?. Mr Everget im really interested, by the way i entirely apreciate this, thank you very much!!!1...

Cevap Gönder

Gunderware PREMİUM · Mar 16, 2020

Hi, I was converting this to version==4 and line 96 gave an error 'n' undefined.
if n < 12
eit := src
eit

Is n the same as i or dcPeriod?

+2 Cevap Gönder

everget WİZARD · Mar 16, 2020

@Gunderware, change it to "bar_index". And use the conversion utility tool!

Cevap Gönder

jaggedsoft PREMİUM · Eki 21, 2018

Thank you for this!
I'm assuming "Rocket Science for Traders" was a good book. Any others you recommend?

+2 Cevap Gönder

everget WİZARD · Eki 22, 2018

@jaggedsoft,
there are some interesting books:
J. Ehlers. Cybernetic Analysis for Stock and Futures"
J. Perl. "DeMark Indicators"
C. Kase "Trading With The Odds"
T. Chande, S. Kroll. "The New Technical Trader"
T. Chande. "Beyond Technical Analysis"
P. Kaufman. "New Trading Systems and Methods"
P. Kaufman. "Smarter Trading"
A. Grimes. "The Art Science of Technical Analysis"

J.M. Hurst. "The Profit Magic of Stock Transaction Timing"
A. Elder. "Trading for a Living. Psychology Trading Tactics Money Management"
A. Elder. "Come Into My Trading Room"

+7 Cevap Gönder

jaggedsoft PREMIUM · Eki 23, 2018

Thank you very much friend. Much appreciated!

Today I learned about Camarilla Points, Demark Pivot Points, Woodies Points, Fibonacci Pivot Points, and Floor Pivots

+2 Cevap Gönder

CoinOperator PRO+ · Mar 4

Unable to create a simple alert on this indicator. Stopped--Calculation Error given. Please advise.

Cevap Gönder