Ehlers Simple Decycler

everget WIZARD Mar 15, 2019

Trend Analysis | Moving Averages | ehlers | decycler | hysteresis | band | highpass | filter | decycle | filtering

2 | 489

This indicator was originally developed by John F. Ehlers (Stocks & Commodities , V.33:10 (September, 2015): "Decyclers").

Mr. Ehlers suggested a way to improve trend identification using high-pass filters. The basic smoothers like SMA , low-pass filters, have considerable lag in their display. Mr. Ehlers applied the high-pass filter and subtracted the high-pass filter output from the time series input. Doing these steps he removed high-frequency short-wavelength components (the ones causing the wiggles) from the time series.

As a result he got a special series of the low-frequency components with virtually no lag - the Decycler.

The Decycler is plotted with two additional lines (the percent-shifts of Decycler) and together they form a hysteresis band.

If the prices are above the upper hysteresis line, then the market is in an **uptrend**. If the prices are below the low hysteresis line, then the market is in a **downtrend**. Prices within the hysteresis band are **trend-neutral**.

Freelance -> Telegram: @alex_everget

A list of Free indicators:
https://bit.ly/2S7EPuN

A list of Paid indicators:
https://bit.ly/33MA81f

Earn $30:
https://www.tradingview.com/gopro/?share_your_love=everget

Website


```pine
//@version=3
// Copyright (c) 2019-present, Alex Orekhov (everget)
// Ehlers Simple Decycler script may be freely distributed under the MIT license.
study("Ehlers Simple Decycler", shorttitle="Decycler", overlay=true)

highpassLength = input(title="Highpass Period", type=integer, defval=125)
upperPercent = input(title="Upper Band % Shift", type=float, minval=0, step=0.1, defval=0.5)
lowerPercent = input(title="Lower Band % Shift", type=float, minval=0, step=0.1, defval=0.5)
src = input(title="Source", type=source, defval=close)
highlightMovements = input(title="Highlight Decycler Movements ?", type=bool, defval=true)

PI = 2 * asin(1)

// High-pass Filter
alphaArg = 2 * PI / (highpassLength * sqrt(2))

alpha = 0.0
alpha := cos(alphaArg) != 0
     ? (cos(alphaArg) + sin(alphaArg) - 1) / cos(alphaArg)
     : nz(alpha[1])

hp = 0.0
hp := pow(1 - (alpha / 2), 2) * (src - 2 * nz(src[1]) + nz(src[2])) + 2 * (1 - alpha) * nz(hp[1]) - pow(1 - alpha, 2) * nz(hp[2])

decycler = src - hp

decyclerColor = highlightMovements ? (decycler >= decycler[1] ? green : red) : #741b47
plot(decycler, title="Decycler", linewidth=2, color=decyclerColor, transp=0)

upperBand = (1 + upperPercent / 100) * decycler
lowerBand = (1 - lowerPercent / 100) * decycler

bandColor = #138484
upperBandPlot = plot(upperBand, title="Upper", color=bandColor)
lowerBandPlot = plot(lowerBand, title="Lower", color=bandColor)
fill(upperBandPlot, lowerBandPlot, title="Background", color=color(#ffd966, 84))
```

Yorumlar

Yararlı veya teşvik edici bir yorum bırakın. Piyasalara birlikte hakim olalım

Alkışlarla yorum | Yorum Paylaş

AlexMayorov · Eki 13, 2019

very nice

+1 Cevap Gönder

everget WİZARD · Eki 13, 2019

@AlexMayorov, spasibo tebe, kamrad!

Cevap Gönder